
Draft is a library for building typed structs with built-in validation support.
Usage
Setup
Add :draft to your project's dependencies in mix.exs:
{:draft, "~> 1.0"}
General Usage
To define a simple Draft struct:
defmodule Book do
  use Draft.Schema

  # Define your struct.
  schema required: true do
    field :id,        :string
    field :title,     :string, min: 1, max: 32
    field :author_id, :string
    field :isbn,      :integer, min: 1_000_000_000, max: 9_999_999_999_999
  end
end
Construction
You can create a struct using new, cast, or from_struct. Only type information is checked during construction.
new
new/1 raises an error for invalid types or missing required fields.
# Using a keyword list
book = Book.new(id: "1", title: "Elixir Draft Tutorial", author_id: "2", isbn: 22222222222)

# Using a map
book = Book.new(%{
  id: "1",
  title: "Elixir Draft Tutorial",
  author_id: "2",
  isbn: 22222222222
})
cast
cast/1 returns a result tuple: {:ok, struct} or {:error, errors}. errors is a keyword list.
{:ok, book} = Book.cast(id: "1", title: "Elixir Draft Tutorial", author_id: "2", isbn: 22222222222)
from_struct
Use from_struct/1 and from_struct!/1 to create a struct from another struct. The bang version raises on errors.
defmodule Document do
  use Draft.Schema

  schema required: true do
    field :id,        :string
    field :title,     :string, min: 1, max: 32
    field :author_id, :string
    field :history,   :list, type: :string, default: []
    field :isbn,      :integer, min: 1_000_000_000, max: 9_999_999_999_999
  end
end

doc = Document.new(id: "1", title: "Elixir Doc", author_id: "2", isbn: 22222222222, history: [])

book = Book.from_struct!(doc)
{:ok, book} = Book.from_struct(doc)

Required Fields
By default, all fields can be nil. Use required: true in the schema to make all fields required.
defmodule Person do
  use Draft.Schema
  schema required: true do
    field :id,    :uuid
    field :name,  :string
    field :age,   :number
    field :amount, :float
  end
end
Fields with default values are considered optional:
field :amount, :float, default: nil
You can make individual fields required:
field :id, :uuid, required: true

Validation
Use Draft.errors(struct) to validate a Draft struct. Errors are returned as a keyword list.
book = Book.new(id: "1", title: "Draft Errors", author_id: "2", isbn: 1)
[isbn: _] = Draft.errors(book)

Inheritance
Draft supports inheritance via the :extends option, including validation rules and types.
defmodule Book do
  use Draft.Schema

  schema required: true do
    field :id,        :string
    field :title,     :string, min: 1, max: 32
    field :author_id, :string
    field :isbn,      :integer, min: 1_000_000_000, max: 9_999_999_999_999
  end
end

defmodule Document do
  use Draft.Schema

  schema extends: Book do
    field :history, :list, type: :string, default: []
  end
end
Multiple inheritance:
defmodule HasAuthor do
  schema required: true do
    field :author_id, :string
  end
end

defmodule HasISBN do
  schema do
    field :isbn, :integer, min: 1_000_000_000, max: 9_999_999_999_999
  end
end

defmodule Book do
  use Draft.Schema

  schema required: true, extends: [HasAuthor, HasISBN] do
    field :id,    :string
    field :title, :string, min: 1, max: 32
  end
end
Overwriting fields:
defmodule Book do
  use Draft.Schema

  schema do
    field :id, :string
  end
end

defmodule Document do
  use Draft.Schema

  schema extends: Book do
    field :id, :uuid, overwrite: true
  end
end

# Invalid UUID
{:error, _} = Document.new(id: "1")

# Valid UUID
doc = Document.new(id: "00000000-0000-0000-0000-000000000000")

Advanced Usage
Map Fields
defmodule Typed do
  use Draft.Schema

  @mapping [
    name:  [:string, length: [min: 5, max: 10]],
    value: [:number, required: false]
  ]

  schema do
    field :stats, :map, fields: @mapping
  end
end
Nested Types
defmodule Book do
  use Draft.Schema

  schema do
    field :title, :string
  end
end

defmodule Library do
  use Draft.Schema

  schema do
    field :books, :list, type: Book, default: []
  end
end
Required Field Validation
field :name, :string, required: true
Length Validation
field :name, :string, length: [min: 2, max: 20]
Pattern Validation
field :password, :string, pattern: ~r/^[[:alnum:]]+$/

Customization
Draft types must implement both Draft.Type.Behaviour and Draft.Validator.Behaviour.
Custom Type (Draft.Type.Behaviour)
defmodule ISBN.Type do
  @behaviour Draft.Type.Behaviour

  def cast(value, _opts) when is_integer(value) and value in 1_000_000_000..999_999_999_999, do:
    {:ok, value}

  def cast(_value, _opts), do:
    {:error, ["value must be a valid ISBN"]}

  def dump(value, _opts), do:
    {:ok, value}
end
Custom Validator (Draft.Validator.Behaviour)
defmodule ISBN.Validator do
  @behaviour Draft.Validator.Behaviour

  def validate(value, opts), do: validate(value, nil, opts)

  def validate(value, _context, _opts), do:
    {:ok, value}

  def validate(_value, _context, _opts), do:
    {:error, ["reason"]}
end
Configuration
In your config/config.exs:
config :types, Draft,
  isbn: ISBN.Type

config :validators, Draft,
  isbn: ISBN.Validator
Usage in Schema
field :isbn_number, :isbn

Built-in Types
	any
	map
	enum
	atom
	uuid
	list
	tuple
	float
	struct
	number
	string
	boolean
	integer
	datetime

Built-in Validators
	inclusion
	exclusion
	required
	length
	format
	number
	fields
	struct
	uuid
	type
	min
	max
	by
	tld
	pattern


TODO
	[ ] Field documentation
	[ ] Validation documentation

The MIT License (MIT)
Copyright (c) 2016 Che Mfoncho (2p4b@proton.me)

Permission is hereby granted, free of charge, to any person obtaining a copy of this software and associated documentation files (the "Software"), to deal in the Software without restriction, including without limitation the rights to use, copy, modify, merge, publish, distribute, sublicense, and/or sell copies of the Software, and to permit persons to whom the Software is furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in all copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE SOFTWARE.

Implementation of this behaviour should be set in validator options as :error_renderer
or in :draft application config with same key.
Common pattern is to expect :message key in validator options and if it not set
use default_message (use get_message function for it).
Result of message function appears in Draft.errors error tuple as last element, for those
validators who use Draft.Validator.ErrorMessage (in general they should).

example
iex> {:ok, _value} = Draft.Type.Datetime.cast("01-01-1900", [])

iex> {:ok, _value} = Draft.Type.Datetime.cast("1900-01-01", [])

iex> {:ok, _value} = Draft.Type.Datetime.cast("2016-02-29T22:25:00-06:00", [])
iex> {:ok, _value} = Draft.Type.Datetime.cast("2023-02-01T14:51:53.468132Z", [])

iex> {:error, _reason} = Draft.Type.Datetime.cast("1900-91-91", [])

Ensure a value is not a member of a list of values.
Options
	:in: The list.
	:message: Optional. A custom error message. May be in EEx format
  and use the fields described in "Custom Error Messages," below.

The list can be provided in place of the keyword list if no other options are needed.
Examples
iex> Draft.Validator.Exclusion.validate(1, [1, 2, 3])
{:error, "must not be one of [1, 2, 3]"}

iex> Draft.Validator.Exclusion.validate(1, [in: [1, 2, 3]])
{:error, "must not be one of [1, 2, 3]"}

iex> Draft.Validator.Exclusion.validate(1, [in: [1, 2, 3], message: "<%= value %> shouldn't be in <%= inspect list %>"])
{:error, "1 shouldn't be in [1, 2, 3]"}

iex> Draft.Validator.Exclusion.validate(4, [1, 2, 3])
{:ok, 4}

iex> Draft.Validator.Exclusion.validate("a", ~w(a b c))
{:error, ~S(must not be one of ["a", "b", "c"])}

iex> Draft.Validator.Exclusion.validate("a", in: ~w(a b c), message: "must not be abc, talkin' 'bout 123")
{:error, "must not be abc, talkin' 'bout 123"}

Ensure a value matches a regular expression.
Options
	:with: The regular expression.
	:message: Optional. A custom error message. May be in EEx format
and use the fields described in "Custom Error Messages," below.

The regular expression can be provided in place of the keyword list if no other options
are needed.
Examples
iex> Draft.Validator.Format.validate("foo", ~r/^f/)
{:ok, "foo"}

iex> Draft.Validator.Format.validate("foo", ~r/o{3,}/)
{:error, "must have the correct format"}

iex> Draft.Validator.Format.validate("foo", [with: ~r/^f/])
{:ok, "foo"}

iex> Draft.Validator.Format.validate("bar", [with: ~r/^f/, message: "must start with an f"])
{:error, "must start with an f"}

iex> Draft.Validator.Format.validate("", [with: ~r/^f/, allow_blank: true])
{:ok, ""}

iex> Draft.Validator.Format.validate(nil, [with: ~r/^f/, allow_nil: true])
{:ok, nil}

Ensure a value is a member of a list of values.
Options
	:in: The list.
	:message: Optional. A custom error message. May be in EEx format
and use the fields described in "Custom Error Messages," below.

The list can be provided in place of the keyword list if no other options are needed.
Examples
iex> Draft.Validator.Inclusion.validate(1, [1, 2, 3])
{:ok, [1,2,3]}

iex> Draft.Validator.Inclusion.validate(1, [in: [1, 2, 3]])
{:ok, [1,2,3]}

iex> Draft.Validator.Inclusion.validate(4, [1, 2, 3])
{:error, "must be one of [1, 2, 3]"}

iex> Draft.Validator.Inclusion.validate("a", ~w(a b c))
{:ok, ~w(a b c)}

iex> Draft.Validator.Inclusion.validate(nil, ~w(a b c))
{:error, ~S(must be one of ["a", "b", "c"])}

iex> Draft.Validator.Inclusion.validate(nil, [in: ~w(a b c), allow_nil: true])
{:ok, nil}

iex> Draft.Validator.Inclusion.validate("", [in: ~w(a b c), allow_blank: true])
{:ok, ""}
Custom Error Messages
Custom error messages (in EEx format), provided as :message, can use the following values:
iex> Draft.Validator.Inclusion.__validator__(:message_fields)
[value: "The bad value", list: "List"]
An example:
iex> Draft.Validator.Inclusion.validate("a", in: [1, 2, 3], message: "<%= inspect value %> is not an allowed value")
{:error, ~S("a" is not an allowed value)}

Ensure a value's length meets a constraint.
Options
At least one of the following must be provided:
	:min: The value is at least this long
	:max: The value is at most this long
	:in: The value's length is within this Range
	:is: The value's length is exactly this amount.

The length for :is can be provided instead of the options keyword list.
The :is is available for readability purposes.
Optional:
	:tokenizer: A function with arity 1 used to split up a
  value for length checking. By default binarys are broken up using
  String.graphemes and all other values (eg, lists) are
  passed through intact. See Draft.Validator.tokens/1.
	:message: Optional. A custom error message. May be in EEx format
  and use the fields described in "Custom Error Messages," below.

Examples
iex> Draft.Validator.Length.validate("foo", 3)
{:ok, "foo"}

iex> Draft.Validator.Length.validate("foo", 2)
{:error, "must have a length of 2"}

iex> Draft.Validator.Length.validate(nil, [is: 2, allow_nil: true])
{:ok, nil}

iex> Draft.Validator.Length.validate("", [is: 2, allow_blank: true])
{:ok, ""}

iex> Draft.Validator.Length.validate("foo", min: 2, max: 8)
{:ok, "foo"}

iex> Draft.Validator.Length.validate("foo", min: 4)
{:error, "must have a length of at least 4"}

iex> Draft.Validator.Length.validate("foo", max: 2)
{:error, "must have a length of no more than 2"}

iex> Draft.Validator.Length.validate("foo", max: 2, message: "must be the right length")
{:error, "must be the right length"}

iex> Draft.Validator.Length.validate("foo", is: 3)
{:ok, "foo"}

iex> Draft.Validator.Length.validate("foo", is: 2)
{:error, "must have a length of 2"}

iex> Draft.Validator.Length.validate("foo", in: 1..6)
{:ok, "foo"}

iex> Draft.Validator.Length.validate("foo", in: 8..10)
{:error, "must have a length between 8 and 10"}

iex> Draft.Validator.Length.validate("four words are here", max: 4, tokenizer: &String.split/1)
{:ok, "foor"}
Custom Error Messages
Custom error messages (in EEx format), provided as :message, can use the following values:
iex> Draft.Validator.Length.__validator__(:message_fields)
[value: "Bad value", tokens: "Tokens from value", size: "Number of tokens", min: "Minimum acceptable value", max: "Maximum acceptable value"]
An example:
iex> Draft.Validator.Length.validate("hello my darling", min: 4, tokenizer: &String.split/1,
...> message: "<%= length tokens %> words isn't enough")
{:error, "3 words isn't enough"}

Ensure a value's value meets a constraint.
Examples
iex> Draft.Validator.Max.validate(3, 3)
{:ok, 3}

iex> Draft.Validator.Max.validate(6, 5)
{:error, "must be less than 5"}

iex> Draft.Validator.Max.validate("foo", 3)
{:ok, "foo"}

iex> Draft.Validator.Max.validate([1, 2, 3, 4], 3)
{:error, "must be less than 3"}

iex> Draft.Validator.Max.validate([1, 2, 3], 3)
{:ok, [1, 2, 3]}

iex> Draft.Validator.Max.validate("foobar", 4)
{:error, "must be less than 4"}

iex> Draft.Validator.Max.validate("foobar", max: 4, message: "must be the right length")
{:error, "must be the right length"}

Ensure a value's value meets a constraint.
Examples
iex> Draft.Validator.Min.validate(3, 3)
{:ok, 3}

iex> Draft.Validator.Min.validate(4, 5)
{:error, "must be greater than 5"}

iex> Draft.Validator.Min.validate("foo", 3)
{:ok, "foo"}

iex> Draft.Validator.Min.validate([1], 3)
{:error, "must be greater than 3"}

iex> Draft.Validator.Min.validate([1, 2, 3], 3)
{:ok, [1, 2, 3]}

iex> Draft.Validator.Min.validate("foo", 4)
{:error, "must be greater than 4"}

iex> Draft.Validator.Min.validate("foo", min: 4, message: "must be the right length")
{:error, "must be the right length"}

Ensure a value is a number.
Options
At least one of the following must be provided:
	:is: The value is a number (integer or float) or not.
	:equal_to: The value is a number equal to this number.
	:greater_than : The value is a number greater than this number.
	:greater_than_or_equal_to: The value is a number greater than or equal to this number.
	:less_than : The value is a number less than this number.
	:less_than_or_equal_to: The value is a number less than or equal to this number.

Optional:
	:message: A custom error message. May be in EEx format and use the fields described
  in Custom Error Messages.
	:allow_nil: A boolean whether to skip this validation for nil values.
	:allow_blank: A boolean whether to skip this validaton for blank values.

The :is option can be provided in place of the keyword list if no other options are set.
When multiple options are than the validator will do an and logic between them.
Examples
Examples when using the :is option:
iex> Draft.Validator.Number.validate("not_a_number", is: true)
{:error, "must be a number"}

iex> Draft.Validator.Number.validate(3.14, is: true)
{:ok, 3.14}

iex> Draft.Validator.Number.validate("not_a_number", is: false)
{:ok, 3.14}

iex> Draft.Validator.Number.validate(3.14, is: false)
{:error, "must not be a number"}
Examples when using the boolean value in options for the :is option:
iex> Draft.Validator.Number.validate("not_a_number", true)
{:error, "must be a number"}

iex> Draft.Validator.Number.validate(3.14, true)
{:ok, 3.14}

iex> Draft.Validator.Number.validate("not_a_number", false)
{:ok, "not_a_number"}

iex> Draft.Validator.Number.validate(3.14, false)
{:error, "must not be a number"}
Examples when using the :equal_to option:
iex> Draft.Validator.Number.validate(3.14, equal_to: 1.41)
{:error, "must be a number equal to 1.41"}

iex> Draft.Validator.Number.validate(3.14, equal_to: 3.14)
{:ok, 3.14}

iex> Draft.Validator.Number.validate(3.14, equal_to: 6.28)
{:error, "must be a number equal to 6.28"}
Examples when using the :greater_than option:
iex> Draft.Validator.Number.validate(3.14, greater_than: 1.41)
{:ok, 3.14}

iex> Draft.Validator.Number.validate(3.14, greater_than: 3.14)
{:error, "must be a number greater than 3.14"}

iex> Draft.Validator.Number.validate(3.14, greater_than: 6.28)
{:error, "must be a number greater than 6.28"}
Examples when using the :greater_than_or_equal_to option:
iex> Draft.Validator.Number.validate(3.14, greater_than_or_equal_to: 1.41)
{:ok, 3.14}

iex> Draft.Validator.Number.validate(3.14, greater_than_or_equal_to: 3.14)
{:ok, 3.14}

iex> Draft.Validator.Number.validate(3.14, greater_than_or_equal_to: 6.28)
{:error, "must be a number greater than or equal to 6.28"}
Examples when using the :less_than option:
iex> Draft.Validator.Number.validate(3.14, less_than: 1.41)
{:error, "must be a number less than 1.41"}

iex> Draft.Validator.Number.validate(3.14, less_than: 3.14)
{:error, "must be a number less than 3.14"}

iex> Draft.Validator.Number.validate(3.14, less_than: 6.28)
{:ok, 3.14}
Examples when using the :less_than_or_equal_to option:
iex> Draft.Validator.Number.validate(3.14, less_than_or_equal_to: 1.41)
{:error, "must be a number less than or equal to 1.41"}

iex> Draft.Validator.Number.validate(3.14, less_than_or_equal_to: 3.14)
{:ok, 3.14}

iex> Draft.Validator.Number.validate(3.14, less_than_or_equal_to: 6.28)
{:ok, 3.14}
Examples when using the combinations of the above options:
iex> Draft.Validator.Number.validate("not_a_number", is: true, greater_than: 0, less_than_or_equal_to: 3.14)
{:error, "must be a number"}

iex> Draft.Validator.Number.validate(0, is: true, greater_than: 0, less_than_or_equal_to: 3.14)
{:error, "must be a number greater than 0"}

iex> Draft.Validator.Number.validate(1.41, is: true, greater_than: 0, less_than_or_equal_to: 3.14)
{:ok, 1.41}

iex> Draft.Validator.Number.validate(3.14, is: true, greater_than: 0, less_than_or_equal_to: 3.14)
{:ok, 3.14}

iex> Draft.Validator.Number.validate(6.28, is: true, greater_than: 0, less_than_or_equal_to: 3.14)
{:error, "must be a number less than or equal to 3.14"}
Custom Error Messages
Custom error messages (in EEx format), provided as :message, can use the following values:
iex> Draft.Validator.Number.__validator__(:message_fields)
[
    value: "Bad value",
    is: "Is number",
    equal_to: "Equal to number",
    greater_than: "Greater than number",
    greater_than_or_equal_to: "Greater than or equal to number",
    less_than: "Less than number",
    less_than_or_equal_to: "Less than or equal to number"
]
An example:
iex> Draft.Validator.Number.validate(3.14, less_than: 1.41,
...> message: "<%= inspect value %> should be less than <%= less_than %>")
{:error, "3.14 should be less than 1.41"}

